

# INFORME COORDINACIÓN HIDROTÉRMICA

Grupo 10

Marzo 2022

Consideraremos el sistema hipotético y los modelos planteados en el problema.

*En todos los modelos ha de satisfacerse la demanda y la reserva rodante térmica horaria, y respetar los mínimos y máximos técnicos de los grupos, así como sus rampas de subida y de bajada, y en los grupos hidráulicos la potencia máxima de turbinación y la ecuación de balance de las reservas, así como sus limitaciones.*

Los índices que vamos a utilizar a lo largo de todos los modelos son:

- $t \in \{ GAL, CAT, MAD, VAL, EXT - AND, CASTL \}$  para los grupos térmicos
- $g \in \{ Tajo, Duero, Sil \}$  para los grupos hidroeléctricos
- $h \in \{ 1, 2, 3, 4 \}$  para cada hora

Los parámetros que necesitaremos son:

- $pmax_t$ : Potencia máxima de cada térmico [MW] /GAL 400, CAT 500, MAD 700, VAL 400, EXT-AND 900, CASTL 800/
- $pmin_t$ : Potencia mínima de cada térmico [MW] /GAL 100, CAT 150, MAD 150, VAL 50, EXT-AND 450, CASTL 200 /
- $rs_t$ : Rampa de subida [Mw por hora] /GAL 200, CAT 300, MAD 500, VAL 300, EXT-AND 600, CASTL 400/
- $rb_t$ : Rampa de bajada [Mw por hora] /GAL 300, CAT 300, MAD 200, VAL 100, EXT-AND 600, CASTL 400/
- $a_t$ : Término independiente [€] /GAL 50, CAT 30, MAD 30, VAL 25, EXT-AND 80, CASTL 70/
- $b_t$ : Término lineal de producción (Coste unitario) [€ por MWh] /GAL 4, CAT 5, MAD 4.2, VAL 4.5, EXT-AND 2, CASTL 3/
- $d_h$ : Demanda por horas [MW] /1 2500 ,2 2800 ,3 3900 ,4 3000/
- $pmax_{turg}$ : potencia maxima [Mw] de la turbina del grupo hidraulico g /Tajo 700,Duero 1500,Sil 600/

- $rmax_g$ : reserva maxima del grupo [Mw] hidraulico g /Tajo 4180000,Duero 6790000,Sil 2600000/
- $rmin_g$ : reserva minima del [Mw] grupo hidraulico g /Tajo 4179000,Duero 6789000,Sil 2598000/
- $RHI_g$ : reserva inicial [Mw] del grupo hidraulico g /Tajo 4179000,Duero 6789000,Sil 2599000/
- $f_g$ : fluyente [Mw] del grupo hidraulico g /Tajo 160,Duero 440,Sil 200/
- $TI_t$ : Término independiente [€] /GAL 400, CAT 450, MAD 500, VAL 200, EXT-AND 600, CASTL 1000/
- $TL_t$ : Término lineal de producción [€ por MWh] /GAL 0.25, CAT 0.2, MAD 0.2, VAL 0.01, EXT-AND 0.1, CASTL 0.2/
- $TC_t$ : Término cuadrático [€MWh<sup>2</sup>] / GAL 0.007, CAT 0.006, MAD 0.0045, VAL 0.009, EXT-AND 0.0015, CASTL 0.002/
- $ca_t$ : Coste de arranque [€] /GAL 10, CAT 20, MAD 10, VAL 15, EXT-AND 20, CASTL 15/
- $cp_t$ : Coste de parada [€] /GAL 5, CAT 10, MAD 5, VAL 10, EXT-AND 15, CASTL 10/

*Modelo 1: Minimizar los costes de producción considerando exclusivamente el término lineal (coste unitario) de la aproximación lineal  $b_i$ . Dar la programación horaria de los grupos, los que están acoplados cada hora, y la generación hidráulica horaria.*

$PT_{th}$   $PH_{gh}$   $Y1_{th}$   $RH_{gh}$   $V_{gh}$  Las variables que necesitamos para este modelo son:

- $PT_{th}$ : variable positiva que determina la producción en MW del grupo térmico  $t$  la hora  $h$ ,
- $PH_{gh}$ : variable positiva que determina la producción en MW del grupo hidráulico  $g$  la hora  $h$ ,
- $Y1_{th}$ : variable binaria que determina si el grupo térmico  $t$  está acoplado a la hora  $h$ ,
- $RH_{gh}$ : variable positiva que determina la reserva en Mwh del grupo  $g$  la hora  $h$ ,
- $V_{gh}$ : vertido en Mwh del grupo  $g$  la hora  $h$ .

A este modelo le corresponde la siguiente función objetivo:

$$Min : \sum_h \sum_t b_t \cdot PT_{th}$$

y las siguientes familias de restricciones:

1.

$$\sum_t PT_{th} + \sum_g PH_{gh} = d_h \quad \forall h$$

que garantiza que se produzca exactamente la demanda para cada hora,

2.

$$PT_{th} \leq pmax_t \cdot Y1_{th} \quad \forall h, t$$

que limita superiormente la potencia de cada grupo térmico cada hora,

3.

$$pmin_t \cdot Y1_{th} \leq PT_{th} \quad \forall h, t$$

que limita inferiormente la potencia de cada grupo térmico cada hora,

4.

$$PT_{th} - PT_{th-1} \leq rs_t \quad \forall h, t$$

que garantiza que se cumpla la rampa máxima de subida de cada grupo térmico cada hora,

5.

$$PT_{th-1} - PT_{th} \leq rb_t \quad \forall h, t$$

que garantiza que se cumpla la rampa máxima de bajada de cada grupo térmico cada hora,

6.

$$PH_{gh} + f_g \leq pmax_{tur_g} \quad \forall h, g$$

que tiene en cuenta la capacidad máxima de las turbinas de cada grupo hidráulico en cada hora,

7.

$$RH_{gh} \leq rmax_g \quad \forall h, g$$

que determina la máxima reserva de cada grupo hidráulico en cada hora,

8.

$$rmin_g \leq RH_{gh} \quad \forall h, g$$

que determina la mínima reserva de cada grupo hidráulico en cada hora,

9.

$$RH_{g1} = RHI_g - PH_{g1} + apor_{g1} - V_{g1} \quad \forall g$$

$$RH_{gh} = RH_{gh-1} - PH_{gh} + apor_{hg} - V_{gh} \quad \forall g, h$$

la ecuación de balance para la primera hora y para las demás, así como

10.

$$\sum_t (pmax_t \cdot Y1_{th} - PT_{th}) \geq 0,2 \cdot d_h \quad \forall h$$

que garantiza que se cumpla la reserva rodante.

El coste eligiendo la disposición óptima de los grupos asciende a 21380.00€. Para ello, cada grupo térmico deberá producir la siguiente cantidad de MW en las horas indicadas.

Generación térmica modelo 1				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
GAL	200.0000	390.0000	170.0000	100.0000
CAT	0	0	0	0
MAD	250.0000	150.0000	150.0000	150.0000
VAL	0	0	0	0
EXT-AND	600.0000	900.0000	900.0000	900.0000
CASTL	400.0000	800.0000	800.0000	800.0000

Las centrales hidroeléctricas deberán producir los MW que indica la siguiente tabla:

Generación hidráulica modelo 1				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
Tajo	190.0000	30.0000	420.0000	180.0000
Duero	460.0000	130.0000	1060.0000	470.0000
Sil	400.0000	400.0000	400.0000	400.0000

*Modelo 2: Asumiendo los costes de producción exclusivamente el término lineal de la aproximación lineal, pero el operador ha de comprar a un precio único unitario pagando a todos el máximo del coste unitario de los grupos que están generando esa hora, minimizar el pago por la energía. Dar la programación horaria de los grupos, los que están acoplados cada hora, la generación hidráulica horaria y el precio horario.*

Para este modelo tenemos que introducir las siguientes variables:

- $casa_h$ : precio de casación en € de la hora  $h$ .

A este modelo le corresponde la siguiente función objetivo:

$$Min : \sum_h d_h \cdot casa_h$$

y habría que añadir las siguientes restricciones sobre  $h$  a las del *Modelo 1*:

$$casa_h \geq b_t \cdot Y1_{th} \quad \forall t$$

que determinará para cada hora  $h$  el precio de casación como el máximo precio de los grupos térmicos acoplados.

Ahora, la función objetivo toma un valor de 50400.00€ . Se observa que este valor es considerablemente superior al del *Modelo 1*. Esto se debe a que ahora todos los MW se pagan al precio de la energía más cara, el precio de casación. De hecho, este modelo es el de mayor coste. Para este coste, la producción de cada grupo térmico deberá ser la siguiente en MW en las horas indicadas.

Generación térmica modelo 2				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
GAL	200.0000	400.0000	300.0000	0
CAT	0	0	0	0
MAD	500.0000	400.0000	200.0000	0
VAL	300.0000	200.0000	200.0000	0
EXT-AND	600.0000	900.0000	900.0000	840.0000
CASTL	400.0000	500.0000	660.0000	260.0000

Las centrales hidroeléctricas deberán producir lo que indica la siguiente tabla:

Generación hidráulica modelo 2				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
Tajo	0	0	280.0000	540.0000
Duero	100.0000	0	1060.0000	960.0000
Sil	400.0000	400.0000	400.0000	400.0000

La disposición óptima de los grupos en este modelo es completamente diferente a la anterior. En el *Modelo 1* no hay costes asociados a los grupos hidráulicos, y por lo tanto estos grupos estaban acoplados siempre. En este modelo se pagan todas las energías al mismo precio, resultando en un menor consumo de energías hidroeléctricas en las 3 primeras horas, y en un mayor consumo de estas en la última hora.

*Modelo 3: Minimizar los costes de producción considerando la aproximación cuadrática e incluyendo costes de arranque y parada. Dar la programación horaria de los grupos, los que están acoplados cada hora, la generación hidráulica horaria y los que arrancan y paran cada hora.*

Para este modelo tenemos que introducir las siguientes variables:

- $Par_{th}$ : variable binaria que determina si se desacopla el grupo térmico  $t$  en la hora  $h$ .
- $Arr_{th}$ : variable binaria que determina si se acopla el grupo térmico  $t$  en la hora  $h$ .

A este modelo le corresponde la siguiente función objetivo:

$$Min : \sum_h \sum_t TC_t \cdot PT_{th} \cdot PT_{th} + TL_t \cdot PT_{th} + TI_t \cdot Y1_{th} + Arr_{th} \cdot cat_t + Par_{th} \cdot cp_t$$

y habría que añadir las siguientes restricciones a las del modelo anterior:

$$Y1_{th} - Y1_{th-1} = Arr_{th} - Par_{th} \quad \forall t, h$$

que determina cuando arranca o para cada grupo térmico.

## Taller de economatemática

---

La función objetivo ahora toma un valor de 20417.50€. Para ello, cada grupo térmico deberá producir la siguiente cantidad de MW en las horas indicadas.

Generación térmica modelo 3				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
GAL	0	0	0	0
CAT	0	0	0	0
MAD	336.9312	336.9312	295.8519	282.3868
VAL	179.0212	179.0212	158.4815	151.7490
EXT-AND	600.0000	900.0000	900.0000	880.4938
CASTL	400.0000	758.0952	665.6667	635.3704

Las centrales hidroeléctricas deberán producir lo que indica la siguiente tabla:

Generación hidráulica modelo 3				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
Tajo	100.0000	0	540.0000	180.0000
Duero	484.0476	225.9524	940.0000	470.0000
Sil	400.0000	400.0000	400.0000	400.0000

Este modelo incluye los costes de arranque y de parada, luego es esperable que no se acoplen o desacoplen muchos grupos térmicos. En efecto, la solución que obtenemos mantiene los los grupos *GAL* Y *CAT* desacoplados las 4 horas, mientras que el resto de grupos térmicos están acoplados siempre.

*Modelo 4: Minimizar los costes de producción considerando la aproximación lineal e incluyendo costes de arranque y parada. Dar la programación horaria de los grupos, los que están acoplados cada hora, los que arrancan y paran cada hora y la generación hidráulica horaria.*

En este modelo utilizaremos las mismas variables y restricciones que en el *Modelo 3*, pero usaremos la siguiente aproximación lineal en la función objetivo.

$$obj = \sum_h \sum_t b_t \cdot PH_{th} + a_t \cdot Y1_{th} + Arr_{th} \cdot ca_t + Par_{th} \cdot cp_t \quad .$$

La función objetivo toma un valor de 22355.00€ . La aproximación (lineal) de costes utilizada en este modelo es menos precisa que la que se utiliza en el modelo anterior (cuadrática), sin embargo el modelo se resuelve más rápido. Para llegar al coste óptimo,

cada grupo térmico deberá producir la siguiente cantidad de MW en las horas indicadas.

Generación térmica modelo 4				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
GAL	200.0000	390.0000	170.0000	100.0000
CAT	0	0	0	0
MAD	250.0000	150.0000	150.0000	150.0000
VAL	0	0	0	0
EXT-AND	600.0000	900.0000	900.0000	900.0000
CASTL	400.0000	800.0000	800.0000	800.0000

Las centrales hidroeléctricas deberán producir lo que indica la siguiente tabla:

Generación hidráulica modelo 4				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
Tajo	190.0000	30.0000	420.0000	180.0000
Duero	460.0000	130.0000	1060.0000	470.0000
Sil	400.0000	400.0000	400.0000	400.0000

Estos valores son los mismos que los obtenidos en el apartado 1. En este caso, tener en cuenta los costes de arranque y de parada, con esta aproximación lineal, no influye en la política óptima.

*Modelo 5: Minimizar los costes de producción considerando la aproximación lineal e incluyendo costes de arranque y parada, si fuera un ciclo cerrado en los grupos térmicos (empiezan como terminan). Dar la programación horaria de los grupos, los que están acoplados cada hora, los que arrancan y paran cada hora y la generación hidráulica horaria.*

Este modelo se formula igual que el Modelo 4, con la salvedad de que hay que añadir las restricciones:

$$PT_{t0} = PT_{t4} \quad \forall t$$

que significa que empezamos y terminamos con la misma producción cada ciclo.

La función objetivo toma un valor de 21205.00 €. El coste aumenta con respecto al modelo anterior porque tenemos menos políticas posibles, al haber introducido la nueva restricción. Para llegar a este coste óptimo, cada grupo térmico deberá producir la siguiente cantidad de MW en las horas indicadas.



Generación térmica modelo 5				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
GAL	0	0	170.0000	100.0000
CAT	0	0	0	0
MAD	150.0000	150.0000	150.0000	150.0000
VAL	0	0	0	0
EXT-AND	900.0000	900.0000	900.0000	900.0000
CASTL	800.0000	790.0000	800.0000	800.0000

Las centrales hidroeléctricas deberán producir lo que indica la siguiente tabla:

Generación hidráulica modelo 5				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
Tajo	0	220.0000	420.0000	180.0000
Duero	250.0000	340.0000	1060.0000	470.0000
Sil	400.0000	400.0000	400.0000	400.0000

*Modelo 6: Minimizar los costes de producción esperados considerando la aproximación lineal e incluyendo costes de arranque y parada (con ciclo abierto, como modelo 4), si hay incertidumbre sobre la demanda de las horas futuras, de modo que:*

- *En la primera hora y en la segunda hora la demanda es conocida (el dato dado)*
- *En la tercera hora, puede ser un 10 % más del dato dado con probabilidad 0,6 y un 15 % menos del dato dado con probabilidad 0,4*
- *En la cuarta hora, puede ser un 10 % más del dato dado con probabilidad 0,6 y un 15 % menos del dato dado con probabilidad 0,4*


*Dar la programación que optimiza el coste esperado de la generación (según el escenario), el valor de la solución estocástica, el valor esperado con información perfecta y el valor esperado de la información perfecta.*

*Opcional: Dar la programación que minimiza el máximo arrepentimiento y comparar*

Se plantea el problema de optimizar el coste esperado de la generación, teniendo en cuenta que la demanda podrá recorrer cuatro caminos posibles. Para ello, tomaremos el *Modelo 4* y lo resolveremos utilizando los datos de los cuatro escenarios posibles. A partir de estos resultados calcularemos el valor esperado con información perfecta. Para el problema estocástico, reformularemos la función objetivo y crearemos nuevas variables que dependan de los escenarios.

La siguiente imagen representa los diferentes escenarios posibles que deberá recoger nuestro

modelo.



Para resolver el modelo determinista, creamos un bucle que recorra los 4 posibles escenarios finales. En cada iteración, la demanda cambiará por la correspondiente a dicho escenario. El resto de parámetros y las variables son las mismas que en el *Modelo 4*. Los resultados han sido:

Para el primer escenario, la función objetivo toma un valor de 24120 €. La programación horaria para los grupos térmicos es la siguiente:

Generación térmica escenario 1				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
GAL	200.0000	312.0000	400.0000	290.0000
CAT	0	0	0	0
MAD	428.0000	228.0000	192.0000	150.0000
VAL	0	0	50.0000	0
EXT-AND	600.0000	900.0000	900.0000	900.0000
CASTL	400.0000	800.0000	800.0000	800.0000

La producción de las centrales hidroeléctricas será la siguiente:

## Taller de economatemática

---

Generación hidráulica escenario 1				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
Tajo	42.0000	0	5488.0000	290.0000
Duero	430.0000	160.0000	1060.0000	470.0000
Sil	400.0000	400.0000	400.0000	400.0000

La demanda de este escenario es superior que la demanda del *Modelo 4* en las horas 3 y 4, por lo tanto, el coste asociado a este escenario también es superior.

Para el segundo escenario, el valor de la función objetivo es de 22438 €. A continuación se muestra la programación horaria correspondiente:

Generación térmica escenario 2				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
GAL	20.0000	0	370.0000	300.0000
CAT	0	0	0	0
MAD	370.0000	170.0000	200.0000	0
VAL	0	0	90.0000	50.0000
EXT-AND	600.0000	900.0000	900.0000	900.0000
CASTL	400.0000	800.0000	800.0000	550.0000

La producción de las centrales hidroeléctricas será la siguiente:

Generación hidráulica escenario 2				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
Tajo	100.0000	0	540.0000	180.0000
Duero	430.0000	160.0000	1060.0000	470.0000
Sil	400.0000	400.0000	400.0000	400.0000

Para el tercer escenario, el valor de la función objetivo es de 21182 €. A continuación se muestra la programación horaria correspondiente:

Generación térmica escenario 3				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
GAL	200.0000	0	135.0000	290.0000
CAT	0	0	0	0
MAD	210.0000	150.0000	150.0000	150.0000
VAL	0	0	0	0
EXT-AND	600.0000	900.0000	900.0000	900.0000
CASTL	400.0000	790.0000	800.0000	800.0000

La producción de las centrales hidroeléctricas será la siguiente:

Generación hidráulica escenario 3				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
Tajo	190.0000	10.0000	440.0000	180.0000
Duero	500.0000	550.0000	490.0000	580.0000
Sil	400.0000	400.0000	400.0000	400.0000

Para el tercer escenario, el valor de la función objetivo es de 18507 €. A continuación se muestra la programación horaria correspondiente:

Generación térmica escenario 4				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
GAL	200.0000	0	0	0
CAT	0	0	0	0
MAD	210.0000	150.0000	150.0000	0
VAL	0	0	50.0000	50.0000
EXT-AND	600.0000	900.0000	900.0000	900.0000
CASTL	400.0000	790.0000	800.0000	525.0000

La producción de las centrales hidroeléctricas será la siguiente:

Generación hidráulica escenario 4				
Hora	Hora 1	Hora 2	Hora 3	Hora 4
Tajo	190.0000	200.0000	225.0000	205.0000
Duero	500.0000	360.0000	790.0000	470.0000
Sil	400.0000	400.0000	400.0000	400.0000

Se puede apreciar que aunque la demanda para las dos primeras horas es la misma en todos los escenarios, la programación cambia en esas horas según el escenario en que nos

encontremos, adaptándose a esos futuros cambios.

El modelo estocástico que vamos a resolver ahora plantea precisamente esto, qué decisiones tomar en las primeras dos horas de manera que se minimice el coste esperado. Para ello, tendremos que tomar como función objetivo la esperanza de la función objetivo definida en el *Modelo 4*.

$$obj = \sum_h \sum_t b_t \cdot PH_{th} + a_t \cdot Y1_{th} + Arr_{th} \cdot ca_t + Par_{th} \cdot cp_t \quad .$$

Se puede ver que ahora vamos a necesitar que las variables dependan también del escenario en que estemos. Por lo tanto, necesitamos redefinir las variables que teníamos en el modelo anterior de la siguiente manera:

$$EPT_{sth}, EPH_{sgh}, EY1_{sth}, ERH_{sgh}, EV_{sgh}, EPar_{sth}, EArr_{sth}, Ecas_{sh}$$

dónde el índice  $s \in \{1, 2, 3, 4\}$  representa los diferentes escenarios. Introduciremos también  $s2$ , un alias de este índice ya que lo necesitaremos para las restricciones del modelo. Las variables son las mismas que las del *Modelo 4*, pero precedidas por una  $E$  (estocásticas) en el nombre y con un índice extra que representa los diferentes escenarios. Además habrá que introducir el siguiente parámetro

- $P_{sh}$ : La probabilidad de que ocurra el escenario  $s$  en la hora  $h$ .

y cambiaremos  $d_h$  (el parámetro que representaba la demanda de cada hora) por  $DEM_{sh}$ , ya que también dependerá del escenario. El resto de parámetros no dependen del escenario, por lo tanto siguen siendo los mismos que antes.

La función objetivo quedaría de la siguiente manera:

$$obj = \sum_h \sum_t \sum_{s \in S_h} P_{sh} \cdot (b_t \cdot EPH_{sth} + a_t \cdot EY1_{sth} + EArr_{sth} \cdot ca_t + EPar_{sth} \cdot cp_t) \quad .$$

donde  $S_h$  son los siguientes conjuntos:  $S_1 = \{1\}$   $S_2 = \{1\}$   $S_3 = \{1, 3\}$   $S_4 = \{1, 2, 3, 4\}$

Para programarlo en GAMS, introducimos la siguiente matriz:

$$LUCI_{sh} = \begin{pmatrix} 1 & 1 & 1 & 1 \\ 1 & 1 & 1 & 2 \\ 1 & 1 & 3 & 3 \\ 1 & 1 & 3 & 4 \end{pmatrix}$$

de forma que la función objetivo quedaría de la siguiente manera:

$$obj = \sum_h \sum_t \sum_{s=LUCI(s,h)} P_{sh} \cdot (b_t \cdot EPH_{sth} + a_t \cdot EY1_{sth} + EArr_{sth} \cdot ca_t + EPar_{sth} \cdot cp_t) \quad .$$

La condición  $s = LUCI(s, h)$  con la matriz definida de la forma anterior, nos ayuda a utilizar

únicamente las variables de los escenarios posibles. Es decir, en las dos primeras horas tenemos un único escenario posible y por lo tanto tendremos una única variable  $EPH_{1th}$  que represente la energía producida por la central  $t$  en la hora  $h$ . En la tercera hora, habrá dos variables, una por cada escenario posible, y en la cuarta habrá cuatro. A continuación describiremos las restricciones que utilizaremos, que son las mismas que el *Modelo 4* pero con las nuevas variables.

1.

$$\sum_t EPT_{ths} + \sum_g EPH_{ghs} = DEM_h \quad \forall h, s | s = LUCI_{sh}$$

que garantiza que se produzca exactamente la demanda para cada hora en cada escenario posible,

2.

$$EPT_{ths} \leq pmax_t \cdot EY1_{ths} \quad \forall h, t, s | s = LUCI_{sh}$$

que limita superiormente la potencia de cada grupo térmico cada hora en cada escenario posible,

3.

$$pmin_t \cdot EY1_{ths} \leq EPT_{ths} \quad \forall h, t, s | s = LUCI_{sh}$$

que limita inferiormente la potencia de cada grupo térmico cada hora en cada escenario posible,

4.

$$EPT_{ths} - EPT_{th-1s2} \leq rs_t \quad \forall t, h | h \neq 1, s | s = LUCI_{sh}, s2 | s2 = LUCI_{sh-1}$$

$$EPT_{t1s} \leq rs_t \quad \forall t, s | s = LUCI_{s1},$$

que garantiza que se cumpla la rampa máxima de subida de cada grupo térmico cada hora y en cada escenario factible<sup>1</sup>,

5.

$$EPT_{th-1s2} - EPT_{ths} \leq rb_t \quad \forall t, h | h \neq 1, s | s = LUCI_{sh}, s2 | s2 = LUCI_{sh-1}$$

que garantiza que se cumpla la rampa máxima de bajada de cada grupo térmico cada hora en cada escenario posible,

---

<sup>1</sup>el índice s2 cumple una condición que garantiza que es escenario anterior a s

6.

$$EPH_{ghs} + f_g \leq pmax_{tur_g} \quad \forall h, g, s | s = LUCI_{sh}$$

que tiene en cuenta la capacidad máxima de las turbinas de cada grupo hidráulico en cada hora en cada escenario posible,

7.

$$ERH_{ghs} \leq rmax_g \quad \forall h, g, s | s = LUCI_{sh}$$

que determina la máxima reserva de cada grupo hidráulico en cada hora en cada escenario posible,

8.

$$rmin_g \leq RH_{ghs} \quad \forall h, g, s | s = LUCI_{sh}$$

que determina la mínima reserva de cada grupo hidráulico en cada hora en cada escenario posible,

9.

$$ERH_{g1} = RHI_g - EPH_{g1s} + apor_{g1} - EV_{g1s} \quad \forall g, s | s = LUCI_{s1}$$

$$ERH_{ghs} = RH_{gh-1} - EPH_{ghs} + apor_{hg} - EV_{ghs} \quad \forall g, h > 1, s | s = LUCI_{sh}$$

la ecuación de balance en cada escenario posible para la primera hora y para las demás, así como

10.

$$\sum_t (pmax_t \cdot EY1_{ths} - EPT_{ths}) \geq 0,2 \cdot DEM_{hs} \quad \forall h$$

que garantiza que se cumpla la reserva rodante, y por último

11.

$$Y1_{ths} - Y1_{th-1s2} = Arr_{ths} - Par_{ths} \quad \forall t, h | h \neq 1, s | s = LUCI_{sh}, s2 | s2 = LUCI_{sh-1}$$

$$Y1_{t1s} = Arr_{t1s} - Par_{t1s} \quad \forall t, s | s = LUCI_{sh}$$

la condición de arranque y parada.

La resolución de este modelo nos lleva a que el valor de la función objetivo es 22622.2880 €. Este coste es mayor al coste del *Modelo 4*, que era 22355.00€. Este resultado es esperable ya que el modelo que tenemos ahora no es determinista, y desconocemos cual será la demanda en algunas etapas, por lo que se prescribe una programación más flexible que se adapte a los escenarios que puedan suceder.

El valor esperado con información perfecta es la media ponderada de los valores que ha tomado la función objetivo en función de la probabilidad de que ocurra cada escenario.

Valor esperado con información perfecta =  $0,36 \cdot 25249,6 + 0,24 \cdot 22438 + 0,24 \cdot 21182 + 0,16 \cdot 18407 = \textit{patatan}$

El valor de información perfecta es la diferencia entre el valor objetivo del modelo estocástico y el valor con información perfecta.

valor de información perfecta =  $22622,288 - \textit{patatan} = \textit{patatatatan}$

#### CONCLUSIONES

Como *patatatatan* es muy pequeño, el modelo estocástico se adapta satisfactoriamente a los distintos escenarios teniendo en cuenta la variabilidad de las demandas y del coste entre los escenarios. Por tanto es recomendable usarlo.



Adjuntamos el código de gams empleado para resolver los modelos:

```
$TITLE **** Grupo 10 Coordinacion Hidrotermica ****

sets
t termicas /GAL, CAT, MAD, VAL, EXT-AND, CASTL/
g hidraulicos /Tajo, Duero, Sil/
h horas /1*4/
s escenarios /1*4/
alias (s,s2);
;

parameters
pmax(t) Potencia máxima de cada término [MW]
/GAL 400, CAT 500, MAD 700, VAL 400, EXT-AND 900, CASTL 800/
pmin(t) Potencia mínima de cada término [MW]
/GAL 100, CAT 150, MAD 150, VAL 50, EXT-AND 450, CASTL 200 /
rs(t) Rampa de subida [Mw por hora]
/GAL 200, CAT 300, MAD 500, VAL 300, EXT-AND 600, CASTL 400/
rb(t) Rampa de bajada [Mw por hora]
/GAL 300, CAT 300, MAD 200, VAL 100, EXT-AND 600, CASTL 400/
a(t) Término independiente [€]
/GAL 50, CAT 30, MAD 30, VAL 25, EXT-AND 80, CASTL 70/
b(t) Término lineal de producción (Coste unitario) [€ por MWh]
/GAL 4, CAT 5, MAD 4.2, VAL 4.5, EXT-AND 2, CASTL 3/
d(h) Demanda por horas [MW]
/1 2500 ,2 2800 ,3 3900 ,4 3000/
pmaxtur(g) potencia maxima [Mw] de la turbina del grupo hidraulico g
/Tajo 700,Duero 1500,Sil 600/
rmax(g) reserva maxima del grupo [Mw] hidraulico g
/Tajo 4180000,Duero 6790000,Sil 2600000/
rmin(g) reserva minima del [Mw] grupo hidraulico g
/Tajo 4179000,Duero 6789000,Sil 2598000/
RHI(g) reserva inicial [Mw] del grupo hidraulico g
/Tajo 4179000,Duero 6789000,Sil 2599000/
f(g) fluyente [Mw] del grupo hidraulico g
/Tajo 160,Duero 440,Sil 200/
TI(t) Término independiente [€]
/GAL 400, CAT 450, MAD 500, VAL 200, EXT-AND 600, CASTL 1000/
TL(t) Término lineal de producción [€ por MWh]
```

```

/GAL 0.25, CAT 0.2, MAD 0.2, VAL 0.01, EXT-AND 0.1, CASTL 0.2/
TC(t) Término cuadrático [€MWh2]
/ GAL 0.007, CAT 0.006, MAD 0.0045, VAL 0.009, EXT-AND 0.0015, CASTL 0.002/
ca(t) Coste de arranque [€]
/GAL 10, CAT 20, MAD 10, VAL 15, EXT-AND 20, CASTL 15/
cp(t) Coste de parada [€]
/GAL 5, CAT 10, MAD 5, VAL 10, EXT-AND 15, CASTL 10/
PROB(s) probabilidad cada escenario [p.u.] /1 0.36, 2 0.24, 3 0.24, 4 0.16/
;
table apor(h,g) aportacion hidraulica [Mwh] de cada rio
    Tajo Duero Sil
1  190  500  220
2  200  550  250
3  250  600  300
4  180  470  200
;
table DEMS(s,h) demanda estocástica [MW] para escenario y hora
    1      2      3      4
1  2500  2800  4290  3300
2  2500  2800  4290  2550
3  2500  2800  3315  3300
4  2500  2800  3315  2550
;
table P(s,h) probabilidad de cada escenario en cada hora
    1      2      3      4
1  1      1      0.6    0.36
2  1      1      0.6    0.24
3  1      1      0.4    0.24
4  1      1      0.4    0.16
;
table LUCI(s,h)
    1      2      3      4
1  1      1      1      1
2  1      1      1      2
3  1      1      3      3
4  1      1      3      4
;
variables

```

obj el valor [€] de la funcion objetivo  
PT(t,h) produccion [MW] del grupo termico t la hora h  
PH(g,h) produccion [MW] del grupo hidraulico g la hora h  
Y1(t,h) esta acoplado el grupo termico t la hora h  
RH(g,h) reserva [Mwh] del grupo g la hora h  
casa(h) precio de casacion [€] la hora h  
V(g,h) vertido [Mwh] del grupo g la hora h  
Par(t,h) el grupo termico t para en la hora h  
Arr(t,h) el grupo termico t arranca en la hora h  
  
EPT(s,t,h) produccion [MW] del grupo termico t la hora h en el modelo estoc  
EPH(s,g,h) produccion [MW] del grupo hidraulico g la hora h en el modelo estoc  
EY1(s,t,h) esta acoplado el grupo termico t la hora h en el modelo estoc  
ERH(s,g,h) reserva [Mwh] del grupo g la hora h en el modelo estoc  
EV(s,g,h) vertido [Mwh] del grupo g la hora h en el modelo estoc  
EPar(s,t,h) el grupo termico t para en la hora h en el modelo estoc  
EArr(s,t,h) el grupo termico t arranca en la hora h en el modelo estoc  
;  
positive variables PT(t,h), PH(g,h), V(g,h),  
EPT(s,t,h), EPH(s,g,h), EV(s,g,h)  
;  
binary variables Y1(t,h), Par(t,h), Arr(t,h),  
EY1(s,t,h), EPar(s,t,h), EArr(s,t,h)  
;  
equations  
fobj1 coste total modelo 1  
fdem demanda  
fpmax potencia maxima  
fpmin potencia minima  
fsubida rampa de subida  
fbajada rampa de bajada  
ffluyente fluyente  
freservamax reserva maxima  
freservamin reserva minima  
fbalance restriccion de balance  
frr reserva rodante  
fobj2 coste total modelo 2  
fcasa el precio de casacion es maximo

```

fobj3 coste total modelo 3
fencendido determina el encendido y apagado de los grupos
fobj4 coste total modelo 4
fc1 garantiza un modelo periodico en lo tocante a rampa de subida
fc2 garantiza un modelo periodico en lo tocante a rampaa de bajada
fc3 garantiza un modelo periodico en lo tocante a encendido

Efdem demanda
Efpmax potencia maxima
Efpmin potencia minima
Efsubida rampa de subida
Efbajada rampa de bajada
Efluyente fluyente
Efreservamax reserva maxima
Efreservamin reserva minima
Efbalance restriccion de balance
Efrr reserva rodanteo
Efobj4 coste total modelo 4
Efencendido determina el encendido y apagado de los grupos
EfbalanceI ecuacion de valance inicial
;
fobj1.. obj =E= sum(h, sum(t, b(t)*PT(t,h)));
fdem(h).. d(h) =E= sum(t, PT(t,h)) + sum(g, PH(g,h));
fpmax(t,h).. PT(t,h) =L= pmax(t)*Y1(t,h);
fpmin(t,h).. pmin(t)*Y1(t,h) =L= PT(t,h);
fsubida(t,h).. PT(t,h)-PT(t,h-1) =L= rs(t);
fbajada(t,h).. PT(t,h-1)-PT(t,h) =L= rb(t);
ffluyente(g,h).. PH(g,h) + f(g) =L= pmxtur(g);
freservamax(g,h).. RH(g,h) =L= rmax(g);
freservamin(g,h).. rmin(g) =L= RH(g,h);
fbalance(g,h).. RH(g,h) =E= RH(g,h-1)$(ord(h)>1) + RHI(g)$(ord(h)=1)
- PH(g,h) + apor(h,g) - V(g,h);
frr(h).. sum(t, pmax(t)*Y1(t,h) - PT(t,h)) =G= 0.2*d(h);
fobj2.. obj =E= sum(h, d(h)*casa(h));
fcasa(t,h).. casa(h) =G= b(t)*Y1(t,h);
fobj3.. obj =E=sum(h, sum(t,TC(t)*PT(t,h)*PT(t,h) + TL(t)*PT(t,h)
+ TI(t)*Y1(t,h)) + sum(t, Arr(t,h)*ca(t) + Par(t,h)*cp(t)) );
fencendido(t,h).. Y1(t,h)-Y1(t,h-1) =E= Arr(t,h) - Par(t,h);

```

```

fobj4.. obj =E= sum(h, sum(t, b(t)*PT(t,h) + a(t)*Y1(t,h)
+ Arr(t,h)*ca(t) + Par(t,h)*cp(t)));
fc1(t,h).. PT(t,h)-PT(t,h--1) =L= rs(t);
fc2(t,h).. PT(t,h--1)-PT(t,h) =L= rb(t);
fc3(t,h).. Y1(t,h)-Y1(t,h--1) =E= Arr(t,h) - Par(t,h);
Efobj4.. obj =E= SUM((s,t,h)$ (ord(s)=LUCI(s,h)), P(s,h)* (a(t)*EY1(s,t,h)
+ b(t)*EPT(s,t,h) + ca(t)*EArr(s,t,h) + cp(t)*Epar(s,t,h)));
Efdem(s,h)$ (ord(s)=LUCI(s,h)).. DEMS(s,h) =E= sum(t, EPT(s,t,h))
+ sum(g, EPH(s,g,h));
Efpmax(s,t,h)$ (ord(s)=LUCI(s,h)).. EPT(s,t,h) =L= pmax(t)*EY1(s,t,h);
Efpmin(s,t,h)$ (ord(s)=LUCI(s,h)).. pmin(t)*EY1(s,t,h) =L= EPT(s,t,h);
Efsubida(s,t,h,s2)$ (ord(s)=LUCI(s,h) and (ord(s2)=LUCI(s,h-1) or ord(h)=1))..
EPT(s,t,h) - EPT(s2,t,h-1) =L= rs(t);
Efbajada(t,h,s,s2)$ (ord(s)=LUCI(s,h) and (ord(s2)=LUCI(s,h-1) or ord(h)=1))..
EPT(s2,t,h-1) - EPT(s,t,h) =L= rb(t);
Effluyente(s,g,h)$ (ord(s)=LUCI(s,h)).. EPH(s,g,h) + f(g) =L= pmaxtur(g);
Efreservamax(s,g,h)$ (ord(s)=LUCI(s,h)).. ERH(s,g,h) =L= rmax(g);
Efreservamin(s,g,h)$ (ord(s)=LUCI(s,h)).. rmin(g) =L= ERH(s,g,h);
Efbalance(s,g,h)$ (ord(h)>1).. ERH(s,g,h) =E= sum(s2$ (ord(s2)=LUCI(s,h-1)),
ERH(s2,g,h-1)) - EPH(s,g,h) +apor(h,g)-EV(s,g,h);
EfbalanceI(s,g,h)$ (ord(h)=1).. ERH(s,g,h) =E= RHI(g) - EPH(s,g,h)
+ apor(h,g) - EV(s,g,h);
Efrr(s,h)$ (ord(s)=LUCI(s,h))..
sum(t, pmax(t)*EY1(s,t,h) - EPT(s,t,h)) =G=0.2*DEMS(s,h);
Efencendido(s,t,h,s2)$ (ord(s)=LUCI(s,h) and (ord(s2)=LUCI(s,h-1) or ord(h)=1))..
EY1(s,t,h)-EY1(s2,t,h-1) =E= EArr(s,t,h) - EPar(s,t,h);

model modelo_1 /fobj1, fdem, fpmax, fpmin, fsubida, fbajada, ffluyente,
freservamax, freservamin, fbalance, frr/;
model modelo_2 /fobj2, fdem, fpmax, fpmin, fsubida, fbajada, ffluyente,
freservamax, freservamin, fbalance, frr, fcase/;
model modelo_3 /fobj3, fdem, fpmax, fpmin, fsubida, fbajada, ffluyente,
freservamax, freservamin, fbalance, frr, fencendido/;
model modelo_4 /fobj4, fdem, fpmax, fpmin, fsubida, fbajada, ffluyente,
freservamax, freservamin, fbalance, frr, fencendido/;
model modelo_5 /fobj4, fdem, fpmax, fpmin, ffluyente,
freservamax, freservamin, fbalance, frr, fencendido/;
solve modelo_1 using MIP minimizing obj;

```

```
solve modelo_2 using MIP minimizing obj;
option MIQCP = cplex;
solve modelo_3 using MIQCP minimizing obj;
solve modelo_4 using MIP minimizing obj;
solve modelo_5 using MIP minimizing obj;
MODEL MODELO_6_DETERM /fobj4, fdem, fpmax, fpmin, ffluyente, fbajada, fsubida,
    freservamax, freservamin, fbalance, frr, fencendido/;
loop(s,
d(h) = DEMS(s,h);
solve MODELO_6_DETERM using MIP minimizing obj;)
MODEL MODELO_6_ESTOC /Efobj4, Efdem, Efpmax, Efpmin, Effluyente, Efsubida,
Efajada, Efreservamax, Efreservamin, Efbalance, Efrr, Efencendido, EfbalanceI/;
solve MODELO_6_ESTOC using MIP minimizing obj;
```